Lecture 19

GPIO
General Purpose Input/Output

Bad joke of the day:

Why are assembly programmers always wet?

Because they work below C-level

Solution to Quiz 5

Task: Write a subroutine that checks whether a given integer n is prime

```
;-------------------------------------------------------------------
; Subroutine: is_prime
; Inputs: unsigned word n in R6 -- returned unchanged
;
; Output: binary value in R13 -- R13 = 1 if n is prime
;                                R13 = 0 if n is composite
;
; All other core registers in R4-R15 unchanged
; Subroutine does not access addressed memory locations
;-------------------------------------------------------------------
```


Approach:

- Check if n is divisible by 2, 3, 4, …, n-1
- If it is divisible by any of these numbers, the number is composite
- If not divisible, the number is prime
- Works only if n>3, the cases n = 0, 1, 2 need separate logic

Solution to Quiz 5 v.1

```
is_prime_1:

            push    R5
            push    R12

            clr.w   R13

            cmp.w   #2, R6
            jlo     ret_from_is_prime_1     ; if n<2 not prime
            jne     larger_than_two         ; if (n>=2 and n!= 2) check n

            mov.w   #1, R13                 ; n=2 is prime
            jmp     ret_from_is_prime_1

larger_than_two:

            mov.w   #2, R5                  ; start checking for divisibility by 2

check_divisibility_1:
            call    #is_divisible
            tst.w   R12                     ; if divisible, then R12==1, n is composite
            jnz     ret_from_is_prime_1

            inc.w   R5                      ; check divisibility by next integer
            cmp.w   R6, R5                  ; until n-1
            jne     check_divisibility_1

; We land here only if the number is prime
            mov.w   #1, R13

ret_from_is_prime_1:
            pop     R12
            pop     R5
            ret
```


Note: `is_divisible` modifies R12
Need to push/pop R12!!

Solution to Quiz 5

What about algorithmic complexity? How many calls to `is_divisible`?

Depends on the nature of the number:

- If n is even, then only one call: n is divisible by 2
- If n is an odd multiple of 3, then two calls: check for divisibility by 2 and 3
- …
- If n is prime, then **n-2** calls to `is_divisible` with 2, 3, …, n-1

How can we speed things up?

- No need to check n>2 if it is an even number `bit.w #BIT0, R6`
- For odd n, it suffices to check divisibility by odd numbers {3, 5, …, n-2}
- For n>4, no need to check for factors > n/2 check only {3, 5, …, floor(n/2)}
- Actually no need to check for factors > sqrt(n)

Solution to Quiz 5 v.2

```
;
; Somewhat improved efficiency
; Test only odd n for divisibility by odd integers in {3, 5, ..., (n-1)/2}
; Note (n+1)/2 > sqrt(n) for all n>=3
; takes quite some time to run
is_prime_2:

            push    R5
            push    R7
            push    R12

            ; n is more likely not prime: Pr(n is prime) ~ 1/ln(n)
            clr.w   R13

            cmp.w   #3, R6                  ; n=3 is prime
            jeq     found_prime
            cmp.w   #2, R6                  ; n=2 is prime
            jeq     found_prime

            jlo     ret_from_is_prime_2     ; if n<2 not prime
            jne     larger_than_two_2       ; if (n>=2 and n!= 2) check n

            mov.w   #1, R13
            jmp     ret_from_is_prime_2

larger_than_two_2:
            ; if n>2 is even it is not prime
            bit.w   #BIT0, R6
            jnc     ret_from_is_prime_2
```

Solution to Quiz 5 v.2

```
            ; test divisibility by odd numbers only
            mov.w   #3, R5
            mov.w   R6, R7              ; R7=n
            inc.w   R7                  ; R7=n+1
            clrc                        ; unsigned divide by 2 using cleared carry bit
            rrc.w   R7                  ; R7=(n+1)/2

check_divisibility:
            ; check if n is divisible by 3, 5, 7, 9, ..., n-1
            call    #is_divisible
            tst.w   R12
            jnz     ret_from_is_prime_2 ; found a factor, n is composite

            incd.w  R5                  ; next odd number
            cmp.w   R7, R5              ; will stop checking at R7 = (n+1)/2
                                        ; this works because both R5 and R6 are odd
            jlo     check_divisibility

; We land here only if the number is prime
found_prime:
            mov.w   #1, R13

ret_from_is_prime_2:
            pop     R12
            pop     R7
            pop     R5
            ret
```


~4 times faster for prime numbers

Learning From Quiz 4

How can we *improve* following code?

Logic is correct

```
69 is_divisible:
70              push    R5
71              push    R6
72
73              clr.w   R12
74
75 check:       cmp.w   R5, R6
76              jlo     found_remainder
77
78              sub.w   R5, R6
79              jhs     check
80
81 found_remainder:
82              tst.w   R6
83              jnz     ret_from_is_divisib
84
85              mov.w   #1, R12
86
87 ret_from_is_divisible:
88              pop     R6
89              pop     R5
90              ret
91
```


No need to push/pop R5 since it is not modified

Learning From Quiz 4

How can we *improve* following code?

Logic is correct

```
67 is_divisible:
68              push    R6
69              clr.w   R12
70
71 check_R6:
72              tst.w   R6
73              jz      yes_divisible
74
75              sub.w   R5, R6
76              cmp.w   R5, R6
77              jlo     ret_is_divisible
78              jz      yes_divisible
79              jhs     check_R6
80
81 yes_divisible:
82              add.w   #1, R12
83              jmp     ret_is_divisible
84
85 ret_is_divisible:
86              pop     R6
87              ret
```


No need: `sub.w` already sets the status bits!!

No need to jump
ret_is_divisible is the next line already!!!

Learning From Quiz 4

How can we *improve* following code?

```
81 not_divisible:
82                  add.w   #0,R12
83                  pop     R6
84                  jmp     ret_to_main
85
86 divisible:
87                  add.w   #1, R12
88                  pop     R6
89                  jmp     ret_to_main
90
91 ret_to_main:
92                  ret
93
```

No need: x + 0 = x

No need to jump to the next line

No need to repeat lines 83-84

```
divisible:
                    add.w   #1, R12

not_divisible:
                    pop     R6
                    ret
```

Learning From Quiz 4

What is the issue with this code?

```
65 is_divisible:
66              push        R6
67              clr.w       R12
68
69              cmp.w       R6, R5
70              jlo         test
71
72 subtract:
73              sub.w       R5, R6
74              jmp         is_divisible
75
76 test:
77              cmp.w       #0, R6
78              jne         subtract
79
80              mov.w       #1, R12
81
82 ret_main:
83              pop         R6
84              ret
85
```

Pushes to stack with each iteration!!!

Pops only once

`(R6) -> PC -> crash`

This is a Mistake

How do we fix it?

```
65 is_divisible:
66              push        R6
67              clr.w       R12
68
69              cmp.w       R6, R5
70              jlo         test
71
72 subtract:
73              sub.w       R5, R6
74              jmp         is_divisible
75
76 test:
77              cmp.w       #0, R6
78              jne         subtract
79
80              mov.w       #1, R12
81
82 ret_main:
83              pop         R6
84              ret
85
```

```
65 is_divisible:
66              push        R6
67              clr.w       R12
68
69 compare:
70              cmp.w       R6, R5
71              jlo         test
72
73 subtract:
74              sub.w       R5, R6
75              jmp         compare
76
77 test:
78              cmp.w       #0, R6
79              jne         subtract
80
81              mov.w       #1, R12
82
83 ret_main:
84              pop         R6
85              ret
```

How MCUs are used in the Real-World

Not the way we have used them so far: we have only used the CPU and memory (RAM/FRAM) of our MCU to do basic data manipulations

When treated like this, the MCU is a very limited computer:

- No real input or interaction with the user/environment
- We defined (hardcoded) data in RAM/FRAM as input to some logic
- Output is limited too: we peek into registers using CCS to view the input

An MCU is intended to do much more

- Interact with the user / environment / other MCUs
- **Input** from buttons, sensors, other MCUs
- **Output** via displays, motors (motor drivers), actuators, control circuitry …

All input / output to the MCU is through **Input/Output Pins**

The MSP430FR6989 Launchpad

This part enables interface to a PC and enables debugging

eZ-FET emulator

Headers with access to selected pins connect I/O devices e.g., sensors, motors, logic analyzer…

MSP430FR6989IPZ

100 Pins

More headers

Only I/O we will use

Push Button S1

Push Button S2

Red LED

Green LED



MSP430FR6989 I/O Options

Ports for GPIO



I/O Through Standard Protocols

The **enhanced Universal Serial Communication Interfaces (eUSCI_A** and **eUSCI_B)** support several standard protocols for I/O

- **Serial Peripheral Interface (SPI)**
- **Inter Integrated Circuit (I²C, I2C, IIC)**
- **Universal Asynchronous Receiver Transmitter (UART)**



How to use?

- Dedicated pins for I/O
- Registers for configuration
- Devices that use these protocols
- e.g., SPI sensors, I²C sensors, I²C motor drivers etc.

All the details in **slau367p.pdf**
Posted to Carmen under Resources

GPIO Ports P1 – P10

Our MCU has 10 **General Purpose Input Output (GPIO) Ports P1 – P10**
(or PA – PJ)

TI refers to these as **Digital I/O ports**

- Each port has **8 pins**
- Pins are labeled as **Px.y** – **x** is the port number, **y** is the pin number

Push buttons and LEDs are connected to GPIO pins



MSP430FR6989IPZ Pinout

slas789d.pdf

80 pins connected to GPIO Ports



Pins are **multiplexed**

MSP430FR6989IPZ Pinout



Pins are **multiplexed**
many are connected to multiple peripherals and need to be configured by selecting one functionality

Green LED

Red LED

Push Buttons S1 & S2

Registers Controlling GPIO Ports

Each port is configured and controlled by a set of **8-bit registers**

Px.y is controlled by **bit y** in the register corresponding to **port x**

e.g.



8-bit Registers

P1SEL0 / P1SEL1
P1SELC
P1DIR
P1OUT P1IN
P1REN
P1IFG
P1IE
P1IES

e.g. output HIGH on p1.0
⇒ set BIT0 in P1OUT
How?

```
bis.b   #BIT0, &P1OUT
```

How not?

```
mov.b   #BIT0, &P1OUT
```

ALWAYS use bit operations
NEVER use move

Addressing GPIO Port Registers

```
bic.b   #BIT0, &P1OUT
bis.b   #BIT0, &P1OUT
```

byte

immediate mode

absolute mode

(rather than symbolic mode)

All addresses are defined in the header file msp430fr69891.h


```
#define P1IN      (PAIN_L)      /* Port 1 Input */
#define P1OUT     (PAOUT_L)     /* Port 1 Output */
#define P1DIR     (PADIR_L)     /* Port 1 Direction */
#define P1REN     (PAREN_L)     /* Port 1 Resistor Enable */
#define P1SEL0    (PASEL0_L)    /* Port 1 Selection 0 */
#define P1SEL1    (PASEL1_L)    /* Port 1 Selection 1 */
#define P1SELC    (PASELC_L)    /* Port 1 Complement Selection */
#define P1IES     (PAIES_L)     /* Port 1 Interrupt Edge Select */
#define P1IE      (PAIE_L)      /* Port 1 Interrupt Enable */
#define P1IFG     (PAIFG_L)     /* Port 1 Interrupt Flag */
```

Registers are replicated for all 10 ports: P2IN, …, P3IN, …, P10IN,…

Configuring Px.y: PxSEL0/ PxSEL1

Function Select Registers: PxSEL0, PxSEL1

Pins are multiplexed



PxSEL0 and PxSEL1 determine the pin function

(PxSELC is a helper register to complement between 00 and 11)

Table 12-2. I/O Function Selection

PxSEL1	PxSEL0	I/O Function
0	0	General purpose I/O is selected
0	1	Primary module function is selected
1	0	Secondary module function is selected
1	1	Tertiary module function is selected

Default values are **PxSEL0.y = 0** and **PxSEL1.y = 0**

⇒ the default function for each pin **Px.y** is **GPIO / Digital I/O**

Configuring Px.y: PxDIR

Direction Register: PxDIR

Selects the **direction** of the corresponding I/O pin: i.e., input or output

PxDIR.y = 0: Pin **Px.y** is switched to **input** direction (Default)

PxDIR.y = 1: Pin **Px.y** is switched to **output** direction

Shorthand notation:

PxDIR.y refers to bit **y** ∈ {0,1, …, 7} of register controlling port **x** ∈ {1, …, 10}

Px.y refers to pin **y** ∈ {0,1, …, 7} of port **x** ∈ {1, …, 10}

Configuring Px.y: PxIN

Input Register: PxIN

Bit **PxIN.y** reflects the value of the intput signal at pin **Px.y**

PxIN.y = 0: Input at pin **Px.y** is LOW

PxIN.y = 1: Input at pin **Px.y** is HIGH

Note: PxIN is a read-only register

You cannot write to it, attempting to write only results in increased current consumption while the write attempt is active

Configuring Px.y: PxOUT – Role 1

Output Register: PxOUT

Bit **PxOUT.y** is the value of the output signal at pin **Px.y**

when the pin is configured as I/O function, **output** direction

PxOUT.y = 0: Output at pin **Px.y** is LOW

PxOUT.y = 1: Output at pin **Px.y** is HIGH

How to write to output?

The red LED is connected to **P1.0**

P1DIR.0 =1 selects the pin as **output**

First set the desired output value, then change the direction
Otherwise, the initial output may be random

```
bis.b   #BIT0, &P1DIR
bis.b   #BIT0, &P1OUT
```

Option 1

```
bis.b   #BIT0, &P1OUT
bis.b   #BIT0, &P1DIR
```

Option 2